

Handreichung
zur Installation von Windows 7-10
in Schulen

Klonen von Windows mit WinPE und ImageX

WinPE ist ein Live-Betriebssystem auf Windows-Basis, das von CD oder USB-Stick gestartet wird. In Verbindung mit einem Imaging-Programm wie ImageX ermöglicht es die Systemsicherung, die Systemwiederherstellung oder das Klonen von Windows-Computern.

INHALT

Systemsicherung und Systemwiederherstellung mit ImageX	3
Einrichten von Festplatten und Partitionen mit Diskpart.....	4
Einrichten des Bootmanagers	6
Erstellen eines Installations-USB-Sticks.....	8
Variation der Datei startjob.bat mit Fehlerprüfung	9
Variation der Datei startjob.bat mit Auswahl des Wim-Images.....	9
Variation der Datei startjob.bat mit grafischer Oberfläche	11
Variation der Datei startjob.bat mit Netzzugriff	12

IMPRESSUM

Akademie für Lehrerfortbildung und Personalführung, Dillingen (<http://alp.dillingen.de>)

Die Handreichung wurde im Rahmen des Projektes SCHULNETZ von Systembetreuern und IT-Multiplikatoren erarbeitet. Sie ist unter der Adresse <http://alp.dillingen.de/schulnetz/materialien> abrufbar.

Dokumentation: Georg Schlagbauer, Akademie Dillingen
Barbara Maier, Akademie Dillingen
Christian Maushart, Landratsamt Dillingen

E-Mail: schlagbauer@alp.dillingen.de

Stand: Mai 2016



ImageX ist ein Befehlszeilenprogramm, mit dem Windows-Images (WIM-Dateien) erstellt oder zurückgespielt werden können. Um ImageX verwenden zu können, wird der Computer vom WinPE-Live-System gestartet.

SYSTEMSICHERUNG UND SYSTEMWIEDERHERSTELLUNG MIT IMAGEX

Anders als die meisten gängigen Imaging-Programme (z. B. DriveSnapshot, Acronis TrueImage, PartImage, etc.) arbeitet ImageX auf Dateiebene und verhält sich dadurch etwas anders als die sektorbasierten Imaging-Programme. Dies hat folgende Auswirkungen:

- Die Größe einer Partition spielt beim Erstellen oder Zurückspielen keine Rolle.
- Die WIM-Datei kann gegebenenfalls auch in der gleichen Partition gespeichert werden, von der das Image erstellt wird.
- Die Partition und das Dateisystem (NTFS) müssen vor dem Zurückspielen einer WIM-Datei vorhanden sein. Wenn ein Image auf einen neuen Computer aufgespielt werden soll, müssen vorher (z. B. mit diskpart) die Partitionen angelegt und formatiert werden.
- Beim Zurückspielen eines Images überschreibt ImageX den Festplatteninhalt nicht vollständig, sondern kopiert den Inhalt der WIM-Datei auf die Festplatte (und überschreibt damit nur vorhandene gleichnamige Dateien).
- Nach dem Zurückspielen eines WIM-Images muss gegebenenfalls noch der Bootloader eingerichtet werden, damit Windows startet.

ERSTELLEN EINES WIM-IMAGES

Der Computer wird vom WinPE-Live-System (CD oder USB-Stick) gestartet. Einfachheitshalber sollte das Programm **imagex.exe** im Live-System enthalten sein.

```
imagex /capture <Quelle> <Ziel> <Beschreibung>
```

```
imagex /capture d: z:\win10.wim "Win10-Grundimage"
```

ZURÜCKSPIELEN EINES WIM-IMAGES

```
imagex /apply <Quelle> <Image-Nr.> <Ziel>
```

```
imagex /apply z:\win10.wim 1 d:
```



EINRICHTEN VON FESTPLATTEN UND PARTITIONEN MIT DISKPART

Diskpart dient zum Einrichten, Partitionieren und zum Formatieren von Festplatten oder USB-Datenträgern.

Mit dem Befehl **diskpart** startet man den interaktiven Modus. Dieser bietet eine relativ gute Hilfefunktion. Wenn Befehle nicht vollständig sind, wird eine Hilfe angeboten. Durch ein vorangestelltes `help` vor einem Befehl können alle Optionen angezeigt werden, ohne dass der Befehl ausgeführt wird.

```
diskpart
DISKPART> help
DISKPART> help create
DISKPART> help create partition
DISKPART> help create partition primary
DISKPART> exit
```

Mit dem `diskpart`-Befehl `list` erhält man einen Überblick über die vorhandene Struktur.

```
diskpart
DISKPART> list
DISKPART> list disk
DISKPART> select disk 0
DISKPART> list partition
DISKPART> select partition 1
DISKPART> ...
```



BEISPIEL ZUM EINRICHTEN EINER FESTPLATTE

Im folgenden Beispiel wird eine Festplatte (disk 0) komplett gelöscht und neu eingerichtet:

- Partition 1 500 MB Systempartition
- Partition 2 50 GB Windows-Startpartition
- Partition 3 Rest Datenpartition

```
diskpart
DISKPART> list disk
DISKPART> select disk 0
DISKPART> clean
DISKPART> create partition primary size=500
DISKPART> create partition primary size=50000
DISKPART> create partition primary
DISKPART> list partition
DISKPART> select partition 1
DISKPART> format fs=ntfs quick label="Hide"
DISKPART> active
DISKPART> select partition 2
DISKPART> format fs=ntfs quick label="Windows 10"
DISKPART> assign
DISKPART> select partition 3
DISKPART> format fs=ntfs quick label="Daten"
DISKPART> assign
DISKPART> exit
```

AUTOMATISIEREN VON DISKPART

Soll zum Einrichten von Computern immer das gleiche Szenario verwendet werden, bietet es sich an, die Befehle von diskpart in einer Textdatei zu notieren und diese per Skript aufzurufen.

```
diskpart /s script.txt
```

Datei skript.txt

```
select disk 0
clean
...
exit
```



EINRICHTEN DES BOOTMANAGERS

Grundsätzlich funktioniert der Startvorgang eines BIOS-basierten Computers wie folgt:

- Der MBR (Master Boot Record) der ersten Festplatte wird gelesen.
- Wurde der MBR nicht verändert (z. B. durch einen Bootmanager wie GRUB), wird in der aktiven Partition der ersten Festplatte nach einem Bootloader gesucht, um das System zu starten.

BOOTMANAGER BOOTMGR

Bis Windows XP wurde dazu der Bootloader NTLDR verwendet, der die Konfigurationsdatei boot.ini interpretierte.

Ab Vista verwendet Windows den Bootmanager bootmgr und das Verzeichnis \Boot. Die eigentlichen Boot-Informationen befinden sich in der Binärdatei \Boot\BCD (Boot Configuration Data). Zum Erstellen des Bootmanagers und zum Verändern der Bootkonfiguration gibt es die Programme bcdboot und bcdedit.

SYSTEMPARTITION UND STARTPARTITION

Bei der Installation von Windows (ab Vista) werden standardmäßig zwei Partitionen angelegt:

- Die **Systempartition** (Je nach Windows-Version 100-500 MB) ist die erste Partition der Festplatte und ist als aktive Partition gekennzeichnet. Sie enthält den Bootmanager.
- Die **Startpartition** oder Windows-Partition enthält die Windows-Betriebssystemdateien, insbesondere das Windows-Verzeichnis.

Datenträger 0 Basis 476,94 GB Online	System-reserviert 500 MB NTFS Fehlerfrei (System, Aktiv, Primäre Partition)	(C:) 476,45 GB NTFS Fehlerfrei (Startpartition, Auslagerungsdatei, Absturzabbild, Primäre Partition)

Abbildung: Partitionierung der Festplatte, aus dem laufenden Windows-Betriebssystem heraus gesehen. Die Systempartition hat keinen Laufwerksbuchstaben und ist deshalb für einen Benutzer nicht erreichbar.

Einen Vorteil hat die Trennung von Systempartition und Startpartition vor allem dann, wenn mehrere Betriebssysteme auf einem Computer installiert sind und der Windows-Bootmanager den Startvorgang dieser Betriebssysteme verwaltet. Ansonsten kann auf die Trennung auch verzichtet werden. Einige Imaging-Programme kommen besser zurecht, wenn sie nur eine Partition sichern müssen.

INSTALLATION DES BOOTMANAGERS

Der Bootmanager kann neu installiert werden, wenn der Computer mit dem WinPE-Livesystem gestartet wurde:

```
bcdboot d:\windows
```

Zu beachten ist folgendes:

- Der Bootmanager wird immer in der aktiven Partition installiert. Die vorgesehene Systempartition muss also die aktive Partition sein.
- Die Laufwerksbuchstaben im WinPE-Live-System und im laufenden Windows-System müssen nicht identisch sein. Unter WinPE wird die aktive Partition normalerweise mit C: angesprochen, die anderen Partitionen erhalten die Laufwerksbuchstaben D:, E:, etc.



ERSTELLEN EINES INSTALLATIONS-USB-STICKS

Ist ein USB-Stick ausreichend schnell und hat genügend Speicherkapazität (ab ca. 8 GB), eignet er sich auch, um neben dem WinPE-Live-System ein vollständiges WIM-Image für eine komplette Windows-Installation aufzunehmen.

Folgende Schritte sind dazu nötig:

- Der USB-Stick wird wie beschrieben mit WinPE vorbereitet.
- Das WIM-Image (im Skript win10.wim) wird auf den USB-Stick kopiert .
- Die zum Vorbereiten der lokalen Festplatte und zum Installieren des Images erforderlichen Befehle werden in die Batch-Datei startjob.bat auf dem USB-Stick geschrieben. Diese Datei startjob.bat wird nach dem Booten des Live-Systems aufgerufen oder – wie im vorherigen Kapitel beschrieben – automatisch gestartet.

DATEI STARTJOB.BAT

```
@echo off
cls
echo Neuinstallation von Windows 10
echo Es wird die gesamte Festplatte dieses Computers gelöscht.
set /p choice=Wollen Sie weitermachen (j/n)?
if not %choice%==j goto Ende

diskpart /s hdprep.txt
imagex /apply win10.wim 1 w:
bcdboot w:\windows

:Ende
```

DATEI HDPREP.TXT

```
select disk 0
clean
create partition primary size=500
create partition primary
select partition 1
format fs=ntfs quick label="Hide"
active
select partition 2
format fs=ntfs quick label="Win10"
assign letter=w
exit
```

Durch die obige Skriptdatei hdprep.txt wird eine eigenständige Bootpartition angelegt.



VARIATION DER DATEI STARTJOB.BAT MIT FEHLERPRÜFUNG

In der nachfolgenden Variation der Datei startjob.bat wird nach jedem Befehl eine Fehlerprüfung durchgeführt. Wenn der Befehl nicht erfolgreich war, wird abgebrochen, ansonsten wird nach einem erfolgreichen Durchlauf der Computer neu gestartet.

```
@echo off
cls
echo Neuinstallation von Windows 10
echo Es wird die gesamte Festplatte dieses Computers geloescht.
set /p choice=Wollen Sie weitermachen (j/n)?
if not %choice%==j goto Ende

diskpart /s hdprep.txt
if errorlevel 1 goto ende

imagex /apply win10.wim 1 w:
if errorlevel 1 goto ende

bcdboot w:\windows
if errorlevel 1 goto ende

exit
:Ende
```

VARIATION DER DATEI STARTJOB.BAT MIT AUSWAHL DES WIM-IMAGES

In der nachfolgenden Variation der Datei startjob.bat wird eine Auswahl aller Wim-Images auf dem USB-Stick geboten. (Dateien mit der Endung wim). Diese Images können zum Klonen ausgewählt werden.

```
@echo off
cls
echo.
echo Neuinstallation von Windows
echo.
echo Folgende Wim-Images stehen zur Auswahl:
echo.

setlocal enabledelayedexpansion
set count=0
for %%i in (*.wim) do (
    set /a count=!count! + 1
    echo Image !count!: %%i
)

echo.
```



```
if %count%==0 (
    echo Es stehen keine Wim-Images zur Auswahl.
    echo.
    goto ende
)

set /p choice=Waehlen Sie ein Image:

set count=0
for %%i in (*.wim) do (
    set /a count=!count! + 1
    if !count!==%choice% set wimimage=%%i
)

if "%wimimage%"==" " (
    echo Es wurde kein Image ausgewaehlt.
    echo.
    goto ende
)

echo.
echo Wim-Image: %wimimage%
echo Es wird die gesamte Festplatte dieses Computers geloescht.
echo.
set /p choice=Wollen Sie weitermachen (j/n)?
if not %choice%==j goto Ende

diskpart /s hdprep.txt
if errorlevel 1 goto ende

imagex /apply %wimimage% 1 w:
if errorlevel 1 goto ende


bcdboot w:\windows
if errorlevel 1 goto ende

rem exit
:Ende
```



VARIATION DER DATEI STARTJOB.BAT MIT GRAFISCHER OBERFLÄCHE

Die Auswahl dies Wim-Images gestaltet sich mit einer grafischen Oberfläche etwas ansprechender. Das folgende Tool „auswaehlen.exe“ ist mit dem Programm AutoIT erstellt. Es bietet die Wim-Dateien im aktuellen Verzeichnis zur Auswahl an. Die Auswahl wird im Temp-Verzeichnis unter imagename.txt gespeichert.



```
@echo off
wim_auswaehlen.exe
if not exist %TEMP%\imagename.txt goto Ende
set /p wimimage= < %TEMP%\imagename.txt

if "%wimimage%"==" " (
    echo Es wurde kein Image ausgewaehlt.
    echo.
    goto ende
)

diskpart /s hdpred.txt
if errorlevel 1 goto ende

imagex /apply %wimimage% 1 w:
if errorlevel 1 goto ende

bcdboot w:\windows
if errorlevel 1 goto Ende

exit
:Ende
```

VARIATION DER DATEI STARTJOB.BAT MIT NETZZUGRIFF

Existiert ein leistungsfähiges Netzwerk (mind. 1 GBit/s), bietet es sich auch an, die Wim-Images nicht auf den USB-Sticks zu sammeln, sondern diese zentral, z. B. auf einer NAS-Box zu pflegen.

```
@echo off
net use Z: /delete 2>NUL
net use Z: \\10.36.104.24\Vorlagen /user:Lehrer Lehrer 1>NUL
if not exist Z:\Wim-Images\wim_auswaehlen.exe goto Ende

set currrdir=%cd%
Z:
cd \Wim-Images

wim_auswaehlen.exe

if not exist %TEMP%\imagenname.txt goto Ende
set /p wimimage= < %TEMP%\imagenname.txt
if "%wimimage%"==" " (
    echo Es wurde kein Image ausgewaehlt.
    echo.
    goto ende
)

diskpart /s %currrdir%\hdprep.txt
if errorlevel 1 goto ende

%currrdir%\imagex /apply %wimimage% 1 w:
if errorlevel 1 goto ende

bcdboot w:\windows
if errorlevel 1 goto Ende

exit
:Ende
```

